

Ellipse Klausur in allgemeiner Lage

Prof. Dr. Dörte Haftendorn: Mathematik mit MuPAD 4 Juli 07 Update 11.07.07

Web: <http://haftendorn.uni-lueneburg.de> www.mathematik-verstehen.de

#####


Ganzzahlige Eigenwerte, Staatsexamen, März 2001

```
keg:=matrix([[5*x^2 + 6*x*y + 5*y^2 -26*sqrt(2)*x
-6*sqrt(2)*y + 74]])
```

$$\left(5 \cdot x^2 + 6 \cdot x \cdot y - \sqrt{2} \cdot x \cdot 26 + 5 \cdot y^2 - \sqrt{2} \cdot y \cdot 6 + 74 \right)$$

```
kegp:=plot::Implicit2d(keg[1]=0,x=0..8,y=-4..1,
GridVisible=TRUE, LineWidth=1):
```

```
plot(kegp,Scaling=Constrained):
```



Dieser Kegelschnitt ist offenbar eine Ellipse.

Durchführung einer Hauptachsentransformation

```
A:= matrix([[5,3],[3,5]]);
a:=matrix([-26*sqrt(2),-6*sqrt(2)]);
at:=linalg::transpose(a);
d:=74;
```

$$\begin{pmatrix} 5 & 3 \\ 3 & 5 \end{pmatrix}$$

$$\left(-26 \cdot \sqrt{2} \quad -6 \cdot \sqrt{2} \right)$$

74

Bestimmung der Achsenrichtungen:

```
evli:=linalg::eigenvectors(A)
```

$$\left[\left[2, 1, \begin{pmatrix} -1 \\ 1 \end{pmatrix} \right], \left[8, 1, \begin{pmatrix} 1 \\ 1 \end{pmatrix} \right] \right]$$

Eigenwerte mit ihrer Vielfachheit und ihrem Eigenvektor.

```
ev1:=-evli[1][3][1]; // negativ genommen,Rechtssystem!
```

```
ev2:=evli[2][3][1];
```

$$\begin{pmatrix} 1 \\ -1 \end{pmatrix}$$

1

$$\begin{pmatrix} 1 \\ 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 \\ 1 \end{pmatrix}$$

```
ew1:=evli[1][1];
```

```
ew2:=evli[2][1];
```

```
2
```

```
8
```

```
linalg::det(ev1.ev2)
```

```
2
```

Wenn diese Determinante negativ ist, tauft man besser die Eigenwerte und Eigenvektoren um. Oder man dreht einen herum, hauptsächlich es wird ein Rechtssystem.

```
float(ev1),float(ev2)
```


$$\begin{pmatrix} 1.0 \\ -1.0 \end{pmatrix}, \begin{pmatrix} 1.0 \\ 1.0 \end{pmatrix}$$

Einzeichnen der Eigenvektoren:

```
ev1p:=plot::Arrow2d(ev1):
```

```
ev2p:=plot::Arrow2d(ev2):
```

```
plot(kegp, ev1p, ev2p, Scaling=Constrained)
```



Wie erwartet sind die Eigenvektoren die Richtungen der Hauptachsen des Kegelschnittes.

Bestimmung der Kegelschnittgleichung

Die Theorie sagt, dass die Eigenwerte nun die Faktoren der quadratischen Glieder werden.

Das heißt hier, dass es zwei quadratische Glieder mit positivem Vorzeichen gibt.

Damit muss es sich um eine Ellipse handeln, wenn es nicht ein entarteter Sonderfall wird.

Da wir oben schon gezeichnet haben, war ja schon klar, dass es eine Ellipse ist.

Soll die gegebene blaue Ellipse in Hauptachsenlage abgebildet werden,

so kann das geschehen durch eine Drehung um den Ursprung, evt. verknüpft mit einer Achsenspiegelung an einer der Koordinatenachsen, gefolgt von einer Verschiebung.

$$\vec{p}' = P^T \vec{p} \quad \text{und} \quad \vec{p}'' = \vec{p}' + \vec{t}$$

Aus den normierten Eigenvektoren erstellt man eine Matrix P.

Die Eigenvektoren stehen gleich von alleine senkrecht aufeinander, wenn die Eigenwerte verschieden sind.

P ist dann eine Orthonormal-Matrix. Für solche ist die inverse Matrix

gleich der transponierten Matrix, bezeichnet mit P^t (oder mit hochgestelltem T). ²

Die Transformationsmatrix, die den Kegelschnitt in eine Lage bewegt, bei der die Hauptachsen parallel

zu den Koordinatenachsen sind, ist P^t.

zu den Koordinatenachsen sind, ist P^t .

Die Theorie sagt nun, dass $P^t \cdot A \cdot P$ die Diagonalmatrix aus den Eigenwerten ist. Davon kann man sich hier überzeugen:

```
ev1n:=linalg::normalize(ev1):
ev2n:=linalg::normalize(ev2):
P:=ev1n.ev2n
```

$$\begin{pmatrix} \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{pmatrix}$$

```
float(P)
```

$$\begin{pmatrix} 0.7071067812 & 0.7071067812 \\ -0.7071067812 & 0.7071067812 \end{pmatrix}$$

```
Pt:=linalg::transpose(P)
```

$$\begin{pmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{pmatrix}$$

```
float(Pt)
```

$$\begin{pmatrix} 0.7071067812 & -0.7071067812 \\ 0.7071067812 & 0.7071067812 \end{pmatrix}$$

Diagonalisierung, die klappt immer, wenn die EV eine ONB bilden, wenn man also so ein P aufstellen kann.

```
Simplify(Pt*A*P);
float(Pt*A*P)
```

$$\begin{pmatrix} 2 & 0 \\ 0 & 8 \end{pmatrix}$$

$$\begin{pmatrix} 2.0 & 0 \\ 0 & 8.0 \end{pmatrix}$$

Evt. winzige 10^{-19} Werte sind numerisch 0.

Vektorschreibweise für die Abbildung $\vec{p} = P \vec{p}'$ und die Quadrikgleichungen, die sich durch Einsetzen ergeben:

$$Q: \vec{p}'^T A \vec{p}' + \vec{a}'^T \vec{p}' + d = 0$$

$$Q: \vec{p}'^T P^T A P \vec{p}' + \vec{a}'^T P \vec{p}' + d = 0$$

$$Q: \vec{p}^T A \vec{p} + \vec{a}^T \vec{p} + d = 0$$

$$Q': \vec{p}'^T P^T A P \vec{p}' + \vec{a}^T P \vec{p}' + d = 0$$

$$Q': \vec{p}'^T D_{EW} \vec{p}' + \vec{a}^T P \vec{p}' + d = 0$$

$$\text{mit } D_{EW} = \begin{pmatrix} \lambda_1 & 0 \\ 0 & \lambda_2 \end{pmatrix}$$

Abbildung der linearen Terme erfordert also

at

$$(-26 \cdot \sqrt{2} \quad -6 \cdot \sqrt{2})$$

atb:=at*P

$$(-20 \quad -32)$$


kegSD:=ew1*x^2+ew2*y^2+atb[1]*x+atb[2]*y+d

$$2 \cdot x^2 - x \cdot 20 + 8 \cdot y^2 - y \cdot 32 + 74$$

kegSDp:=plot::Implicit2d(kegSD=0,x=-3..7,y=-4..3,

LineWidth=1,LineColor=[0,1,0],GridVisible=TRUE):

plot(kegp,kegSDp, Scaling=Constrained)



Bei der Betrachtung dieses Bildes fällt auf, dass es sich um eine positive Drehung um den Ursprung handelt.

Pt, float(Pt);

Dr:=phi->matrix([[cos(phi), -sin(phi)], [sin(phi), cos(phi)]]): Dr(`ϕ`);

$$\begin{pmatrix} \frac{\sqrt{2}}{2} & -\frac{\sqrt{2}}{2} \\ \frac{\sqrt{2}}{2} & \frac{\sqrt{2}}{2} \end{pmatrix}, \begin{pmatrix} 0.7071067812 & -0.7071067812 \\ 0.7071067812 & 0.7071067812 \end{pmatrix}$$

4

$$\begin{pmatrix} \cos(\varphi) & -\sin(\varphi) \\ \sin(\varphi) & \cos(\varphi) \end{pmatrix}$$

$$\begin{pmatrix} \cos(\varphi) & -\sin(\varphi) \\ \sin(\varphi) & \cos(\varphi) \end{pmatrix}$$

Der Vergleich mit der allgemeinen Drehmatrix ergibt, dass es mit dem Anpassen von phi klappen kann:

```
cos(phi)=float(Pt[1,1]);sin(phi)=float(Pt[2,1]);
```

$$\cos(\text{phi}) = 0.7071067812$$

$$\sin(\text{phi}) = 0.7071067812$$

```
arccos(Pt[1,1]);
```

```
wi:=arcsin(Pt[2,1]);
```

$$\frac{\pi}{4}$$

$$\frac{\pi}{4}$$

Daraus folgt umgerechnet ins Winkelmaß:

```
float(arccos(Pt[1,1])/PI*180);
```

```
wig:=float(arcsin(Pt[2,1])/PI*180);
```

$$45.0$$

$$45.0$$

Damit kann die gemeinsame Lösung nur der negative Winkel wi sein.

So passt es auch zur Zeichnung.

Einzeichnen des Zwischenbildes:

```
p:=matrix([x,y]);
```

```
pt:=linalg::transpose(p);
```

$$\begin{pmatrix} x \\ y \end{pmatrix}$$

$$(x \ y)$$

```
float(Dr(wi)),float(Pt)
```

$$\begin{pmatrix} 0.7071067812 & -0.7071067812 \\ 0.7071067812 & 0.7071067812 \end{pmatrix}, \begin{pmatrix} 0.7071067812 & -0.7071067812 \\ 0.7071067812 & 0.7071067812 \end{pmatrix}$$

Beachte, dass die inverse Drehung die ist, die um den negativen Winkel dreht.

```
kegD:=simplify(pt*Dr(wi)*A*Dr(-wi)*p+at*Dr(-wi)*p+d);
```

$$\left(2 \cdot x^2 - x \cdot 20 + 8 \cdot y^2 - y \cdot 32 + 74 \right)$$

Dieses hat man ja auch als transformierte Gleichung erhalten.


```
kegSD
```

$$2 \cdot x^2 - x \cdot 20 + 8 \cdot y^2 - y \cdot 32 + 74$$

5

```
kegDp:=plot::Implicit2d(kegD[1],x=-1..7,y=-2..2,
```

```
LineWidth=1,LineColor=[1,0,1],LineStyle=Dashed,GridVisible=TRUE):
plot(kegp,kegSDp,kegDp,Scaling=Constrained)
```



```
#####
```

Bestimmung der Translation:

```
kegSD
```

$$2 \cdot x^2 - x \cdot 20 + 8 \cdot y^2 - y \cdot 32 + 74$$

Die Arbeitsweise ist dieselbe wie bei der Herstellung der Scheitelform einer Parabel. Hier durch Hinsehen:

```
xterm:=hold(2*(x-5)^2);expand(xterm);
```

```
yterm:=hold(8*(y-2)^2);expand(yterm);
```

$$2 \cdot (x - 5)^2$$

$$2 \cdot x^2 - x \cdot 20 + 50$$

$$8 \cdot (y - 2)^2$$

$$8 \cdot y^2 - y \cdot 32 + 32$$

```
dd:=-50-32+d
```

$$-8$$

Also

```
kegSDK:=xterm+yterm+dd
```

$$2 \cdot (x - 5)^2 + 8 \cdot (y - 2)^2 - 8$$

```
expand(kegSDK) //Probe
```

$$2 \cdot x^2 - x \cdot 20 + 8 \cdot y^2 - y \cdot 32 + 74$$

passt.

Letzter Teil der Hauptachsentransformation ist die Translation t

```
t:=matrix([-5,-2])
```

$$\begin{pmatrix} -5 \\ -2 \end{pmatrix}$$

6

$$\vec{p}'' = \vec{p}' + \vec{t} \quad \text{also} \quad \vec{p}' = \vec{p}'' - \vec{t} \quad \text{Das ergibt:}$$

$$\vec{p}'' = \vec{p}' + \vec{t} \quad \text{also} \quad \vec{p}' = \vec{p}'' - \vec{t} \quad \text{Das ergibt:}$$

```
kegH:=ew1*x^2+ew2*y^2+dd
```


$$2 \cdot x^2 + 8 \cdot y^2 - 8$$

Angabe der Gleichung in der üblichen Form:

```
hold(x^2/4+y^2=1)
```

$$\frac{x^2}{4} + y^2 = 1$$

```
kegHp:=plot::Implicit2d(kegH,x=-3..4,y=-2..3,
LineWidth=1,LineColor=[1,0,0],GridVisible=TRUE):
plot(kegp,kegSDp,kegDp,kegHp,Scaling=Constrained)
```



Bestimmung des ursprünglichen Scheitelpunktes:

$$\vec{m}'' = \begin{pmatrix} 0 \\ 0 \end{pmatrix}, \quad \vec{m}' = -\vec{t}, \quad \vec{m} = P(-\vec{t})$$

```
m:=P*(-t);
```

```
float(m)
```

$$\begin{pmatrix} \frac{7 \cdot \sqrt{2}}{2} \\ -\frac{3 \cdot \sqrt{2}}{2} \end{pmatrix}$$

$$\begin{pmatrix} 4.949747468 \\ -2.121320344 \end{pmatrix}$$

Bestimmung des Urbildes des eines Punktes der Parabelachse:

$$\vec{r}'' = \begin{pmatrix} r \\ 0 \end{pmatrix}, \quad \vec{r}' = \begin{pmatrix} r \\ 0 \end{pmatrix} - \vec{t}, \quad \vec{r} = P \begin{pmatrix} r \\ 0 \end{pmatrix} + \vec{m}, \quad r = r \cdot \overrightarrow{ev_1} + \vec{m}$$

Mit ev1 rechts ist hier der normierte 1. Eigenvektor gemeint.P

```
r:=2: //auf der x-Achse
```

```
rur:=r*ev1n+m;
```

7

$$\begin{pmatrix} \frac{9 \cdot \sqrt{2}}{2} \\ -\frac{5 \cdot \sqrt{2}}{2} \end{pmatrix}$$

$$\begin{pmatrix} \frac{9 \cdot \sqrt{2}}{2} \\ -\frac{5 \cdot \sqrt{2}}{2} \end{pmatrix}$$

```

mp:=plot::Point2d(m, PointColor=[0,1,1]):
msp:=plot::Point2d(-t, PointColor=[0,1,1]):
rurp:=plot::Point2d(rur):
rp:=plot::Point2d([r,0]):
evlurp:=plot::Arrow2d(m,m+ev1):
ev2urp:=plot::Arrow2d(m,m+ev2):
tp:=plot::Arrow2d(-t, [0,0]):
plot(kegp,kegSDp,kegDp,kegHp, mp,msp,rurp,rp,tp,
     evlurp,ev2urp,PointSize=2,Scaling=Constrained)

```

